

HP-41C External Trigger for HP-3468A by Fred Lipschultz  
 PPC Calculator Journal Volume 9 No 7 Pg 12 Oct-Nov 1982  
 Program requires HP 82160A HPIL Module & HP 3468A Meter

```

01 LBL "REFCHNL" ;Reference Channel Routine
02 XROM 28,27 ;AUTOIO - Set interface to AUTO mode
03 SREG 06 ;SigmaReg 06 sets statistics regs 06-11
04 "6 MAR 1982"
05 AVIEW
06 "QUARTZ M993"
07 AVIEW
08 "LASER BEAM"
09 AVIEW
10 "HP3468A"
11 XROM 28,28 ;FINDID - Find loop identifier of HP3468A
12 XROM 28,39 ;SELECT - Select HP3468A for communication
13 XROM 28,38 ;REMOTE - Set HP3468A to remote mode
14 "24 TIME-HR.MIN"
15 AVIEW ;Enter current time in 24 hour mode as HR.MIN
16 STOP ;Wait for user to enter HH.MM, press R/S
17 CLA
18 FIX 0
19 ENTER
20 INT
21 ARCL X
22 ">HR "
23 RDN
24 FRC
25 100
26 *
27 ARCL X
28 ">MIN"
29 AVIEW ;Display ##HR ##MIN current time message
30 CLS ;CLSigma - Clear statistics registers
31 "F1N5T2" ;DC Volts, 5.5 digits, single trigger
32 XROM 28,35 ;OUTA - send setup string to HP3468A
33 ENG 5 ;5 digit engineering mode for voltage
34 .009 ;Loop counter ii.fff = 0.009
35 STO 00 ;Setup for 10 reference readings
36 CLST ;Clear stack before reading voltages
37 "REF CHNL ONLY"
38 AVIEW ;Request reference channel voltage
39 STOP ;Wait for user to press R/S
40 LBL 01 ;Acquire reference channel voltage
41 XROM 28,41 ;TRIGGER - request a reading from HP3468A
42 XROM 28,30 ;IND - input the reading from HP3468A
43 S+ ;Sigma+
44 ISG 00 ;Increment loop counter
45 GTO 01 ;Get next reference channel reading
46 MEAN ;Calculate mean of 10 readings
47 STO 00 ;Store mean in register 00
48 "REF CH MEAN= "
49 ARCL X
50 AVIEW ;Display reference channel mean voltage
51 SDEV ;Calculate standard deviation of voltage
52 X^2
53 "VAR.= "
54 ARCL X
55 AVIEW ;Display reference channel variance
56 "D2RESET ATTN" ;D2 = Text Display Mode of HP3468A
57 XROM 28,35 ;OUTA - Send setup string to HP3468A
58 AVIEW ;Display text message on HP-41C screen
59 STOP ;Wait for user to reset meter and press R/S
60 "D1" ;D1 = Normal Display Mode of HP3468A
61 XROM 28,35 ;OUTA - send setup string to HP3468A
62 LBL "EXTTRG2" ;External Trigger Routine
63 "NO. OF RDGS?"
64 PROMPT ;Request number of readings to be taken
65 STO 01 ;Store number of readings in register 01
66 "DEL T?<SEC>"

```

```

67 PROMPT ;Request time interval for readings
68 STO 02 ;Store time interval in register 02
69 15
70 RCL 01
71 1
72 -
73 RCL Y
74 +
75 1 E3
76 / ;Loop counter ii.fff where ii = 15,
77 + ;fff = 15 + number of readings - 1.
78 STO 03 ;Store loop counter ii.fff in register 03
79 LBL 30 ;Set meter to read DC current trigger
80 "F5R1N3" ;DC Amps, 3 amp range, 3.5 digits
81 XROM 28,35 ;OUTA - send setup string to HP3468A
82 LBL 02 ;Busy waiting loop for current trigger
83 XROM 28,41 ;TRIGGER - request a reading from HP3468A
84 XROM 28,30 ;IND - input the reading from HP3468A
85 X>0? ;While current is > 0 amps, keep looping
86 GTO 02 ;0 or negative current will trigger reading
87 "F1R2RAN5T2" ;DC Volts, 2-3 V range, auto, 5.5 digits, single trigger
88 XROM 28,35 ;OUTA - send setup string to HP3468A
89 XROM 28,41 ;TRIGGER - request a reading from HP3468A
90 XROM 28,30 ;IND - input the reading from HP3468A
91 STO IND 03 ;Store the reading in a register
92 ISG 03 ;Increment to point to next register
93 GTO 30 ;Loop to reset trigger for next reading
94 "DONE" ;Message when finished reading
95 BEEP ;Audible signal after last reading
96 PROMPT ;Corrected by JFW, was "PROMPT" in V9N7P12
97 LBL "PRTDTA" ;Print Data Routine
98 FIX 0 ;0 digit fixed display mode for number of seconds
99 "TIME: "
100 ARCL 02
101 ">" SEC"
102 AVIEW ;Show TIME: <time interval for readings> SEC
103 ENG 5 ;5 digit engineering display mode for voltage
104 "REF CHNL= "
105 ARCL 00
106 ">" VDC"
107 AVIEW ;Show REF CHNL= <reference channel mean> VDC
108 15
109 RCL 01
110 1
111 -
112 RCL Y
113 +
114 1 E3
115 / ;Loop counter ii.fff where ii = 15,
116 + ;fff = 15 + number of readings - 1.
117 STO 03 ;Store loop counter ii.fff in register 03
118 1
119 ENTER
120 ENTER
121 ENTER ;Fill stack with 1 for reading number
122 LBL 08 ;Display voltage readings
123 FIX 0 ;0 digit fixed display mode for reading number
124 CLA ;Clear ALPHA register before composing message
125 ARCL X ;Get reading number from X register into ALPHA
126 ">": " ;Append 1 colon, 2 spaces
127 ENG 5 ;5 digit engineering display mode for voltage
128 ARCL IND 03 ;Get voltage reading
129 AVIEW ;Display <reading number>: <voltage reading>
130 + ;Increment reading number
131 ISG 03 ;Increment loop counter
132 GTO 08 ;Print next voltage reading
133 END

```

HP-41C External Trigger for HP-3468A by Fred Lipschultz  
PPC Calculator Journal Volume 9 No 7 Pg 12 Oct-Nov 1982  
Program requires HP 82160A HPIL Module & HP 3468A Meter

0192C000F80052454643484E4CA71B9906FA36204D41522031  
3938327EFB51554152545A204D3939337EFA4C415345522042  
45414D7EF748503334363841A71CA727A726FE32342054494D  
452D48522E4D494E7E84879C0083689B73F47F485220756911  
1010429B73F47F4D494E7E70F646314E355432A7239E051A10  
10193073FD5245462043484E4C204F4E4C597E8402A729A71E  
479600B2007C30FD524546204348204D45414E3D209B737E7D  
51F65641522E3D209B737EFC44325245534554204154544EA7  
237E84F24431A723C000F80045585454524732FC4E4F2E204F  
4620524447533F8E31FB44454C20543F3C5345433E8E321115  
211141907240111B13434033CF1EF6463552314E33A72303A7  
29A71E64B300FA4631523252414E355432A723A729A71E9183  
9603D0001EF4444F4E45868EC000F7005052544454419C00F6  
54494D453A209B02F57F205345437E9E05FA5245462043484E  
4C3D209B00F57F205644437E1115211141907240111B134340  
3311838383099C00879B73F47F3A20209E059B837E409603B9  
00C0000DDC


402 BYTES

Program Registers Needed: 58

Row 1 (1 - 2)



Row 2 (3 - 4)



Row 3 (5 - 6)



Row 4 (7 - 9)




Row 5 (10 - 13)



Row 6 (13 - 14)



Row 7 (14 - 22)



Row 8 (22 - 28)



Row 9 (28 - 31)



Row 10 (32 - 37)



Row 11 (37 - 39)



Row 12 (40 - 48)



Row 13 ( 48 )



Row 14 (49 - 54)



Row 15 (54 - 56)



Row 16 (56 - 62)



# HP-41C Ext. Trig. HP-3468A by Fred Lipschultz PPC V9 N7 P12 Oct-Nov 1982

Row 17 (62 - 63)



Row 18 (63 - 66)



Row 19 (66 - 69)



Row 20 (69 - 78)



Row 21 (79 - 83)



Row 22 (83 - 87)



Row 23 (87 - 92)



Row 24 (92 - 97)



Row 25 (97 - 99)



Row 26 (99 - 102)



Row 27 (103 - 104)



Row 28 (105 - 110)



Row 29 (111 - 120)



Row 30 (121 - 127)



Row 31 (127 - 133)

